Pensieve header: The full sl_2 invariant using the Drinfel'd double. Based on Projects/SL2Invariant/SL2Invariant.nb.

Program

Program

Utilities

```
In[∘]:= $k = 2; (*ℏ=γ=1;*)
```

Canonical Form:

Program

```
In[∘]:= CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[
    Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
```

The Kronecker δ:

Program

```
In[∘]:= Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

```
In[∘]:= 𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Variables and their duals:

Program

```
In[∘]:= {t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

Finite Zips:

Program

```
In[∘]:= collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] :=
  (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

Program

```
In[•]:= QZip_ζs_List@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → qt.(zs + ys)];
    Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    CF /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zip_ζs[e^Q1 (P /. zrule)]] ];
```

Upper to lower and lower to Upper:

Program

```
In[•]:= U2l = {B_i_^p_. → e^(-p ħ γ b_i), B^p_. → e^(-p ħ γ b), T_i_^p_. → e^(p ħ t_i), T^p_. → e^(p ħ t), 𝒜_i_^p_. → e^(p γ α_i), 𝒜^p_. → e^(p γ α)};
l2U = {e^(c_. b_i_+d_.) :→ B_i^(-c/(ħ γ)) e^d, e^(c_. b+d_.) :→ B^(-c/(ħ γ)) e^d,
    e^(c_. t_i_+d_.) :→ T_i^(c/ħ) e^d, e^(c_. t+d_.) :→ T^(c/ħ) e^d,
    e^(c_. α_i_+d_.) :→ 𝒜_i^(c/γ) e^d, e^(c_. α+d_.) :→ 𝒜^(c/γ) e^d,
    e^ℰ_ :→ e^Expand@ℰ};
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^{Q} as a single"P". Here the z's are b and α and the ζ's are β and a.

Program

```
In[•]:= LZip_ζs_List@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    CF /@ 𝔼[L2, Q2, Det[lt] e^(-L2-Q2) Zip_ζs[e^(L1+Q1) (P /. U2l /. zrule)]] //. l2U ];
```

Program

```
In[•]:= B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := Module[{n}, Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_n@i, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_n@i, {i, {is}}]
    ] // LZip_Join@@Table[{β_n@i, τ_n@i, a_n@i}, {i, {is}}] // QZip_Join@@Table[{ξ_n@i, y_n@i}, {i, {is}}] ];
B_is___[L_, R_] := B_{is}[L, R];
```

Program

𝔼 morphisms with domain and range.

Program

In[∘]:=
```
B_is_List[𝔼_d1_→r1_[L1_, Q1_, P1_], 𝔼_d2_→r2_[L2_, Q2_, P2_]] :=
  𝔼_(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is]) @@ B_is[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_d1_→r1_[L1_, Q1_, P1_] // 𝔼_d2_→r2_[L2_, Q2_, P2_] :=
  B_r1⋂d2[𝔼_d1→r1[L1, Q1, P1], 𝔼_d2→r2[L2, Q2, P2]];
𝔼_d1_→r1_[L1_, Q1_, P1_] ≡ 𝔼_d2_→r2_[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_d1_→r1_[L1_, Q1_, P1_] 𝔼_d2_→r2_[L2_, Q2_, P2_] ^:=
  𝔼_(d1⋃d2)→(r1⋃r2) @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_d_→r_[L_, Q_, P_]_$k_ := 𝔼_d→r @@ 𝔼[L, Q, P]_$k;
𝔼_[ℰ___][i_] := {ℰ}〚i〛;
```

Program

"Define" code

Define[lhs = rhs,   ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

In[∘]:=
```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_is__ = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
   ReleaseHold[Hold[
      SD[op_nisp,$k_Integer, Block[{i, j, k}, op_isp,$k = ℰ; op_nis,$k]];
      SD[op_isp, op_{is},$k]; SD[op_sis__, op_{sis}];
     ] /. {SD → SetDelayed,
      isp → {is} /. {i → i_, j → j_, k → k_},
      nis → {is} /. {i → ii, j → jj, k → kk},
      nisp → {is} /. {i → ii_, j → jj_, k → kk_}
     }] ]]
```

Program

The Fundamental Tensors

Program

In[∘]:=
```
Define[am_i,j→k = 𝔼_{i,j}→{k}[(α_i + α_j) a_k, (e^(-γ α_j) ξ_i + ξ_j) x_k, 1]_$k,
 bm_i,j→k = 𝔼_{i,j}→{k}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^((e^(-ϵ β_i) - 1) η_j y_k)]_$k]
```

Program

In[∘]:=
```
Define[R_i,j =
  𝔼_{}→{i,j}[ħ a_j b_i, ħ x_j y_i, e^(Sum[(1 - e^(γ ϵ ħ))^k (ħ y_i x_j)^k / (k (1 - e^(k γ ϵ ħ))), {k, 2, $k+1}])]_$k]
```

$$R_{i,j} = \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar a_j b_i,\ \hbar x_j y_i,\ e^{\wedge}\left(\sum_{k=2}^{\$k+1} \frac{\left(1-e^{\gamma\epsilon\hbar}\right)^k (\hbar y_i x_j)^k}{k\left(1-e^{k\gamma\epsilon\hbar}\right)}\right)\right]_{\$k}$$

Program

In[∘]:=
```
Define[R̄i,j = 𝔼{}→{i,j}[-ℏ aj bi, -ℏ xj yi / Bi,
    1 + If[$k == 0, 0, (R̄{i,j},$k-1)$k[3] -
       (((R̄{i,j},0)$k R1,2 (R̄{3,4},$k-1)$k) // (bmi,1→i amj,2→j) // (bmi,3→i amj,4→j)) [3]]],
  Pi,j = 𝔼{i,j}→{}[βi αj / ℏ, ηi ξj / ℏ, 1 + If[$k == 0, 0, (P{i,j},$k-1)$k[3] -
       (R1,2 // ((P{1,j},0)$k (P{i,2},$k-1)$k)) [3]]]]
```

Program

In[∘]:=
```
Define[aSj = R̄i,j ~ Bi ~ Pi,j,
  a̅S̅i = 𝔼{i}→{i}[-ai αi, -xi 𝒜i ξi, 1 + If[$k == 0, 0, (a̅S̅{i},$k-1)$k[3] -
       ((a̅S̅{i},0)$k ~ Bi ~ aSi ~ Bi ~ (a̅S̅{i},$k-1)$k) [3]]]]
```

Program

In[∘]:=
```
Define[bSi = Ri,1 ~ B1 ~ aS1 ~ B1 ~ Pi,1,
  b̅S̅i = Ri,1 ~ B1 ~ a̅S̅1 ~ B1 ~ Pi,1,
  aΔi→j,k = (R1,j R2,k) // bm1,2→3 // P3,i,
  bΔi→j,k = (Rj,1 Rk,2) // am1,2→3 // Pi,3]
```

Program

In[∘]:=
```
Define[dmi,j→k = (𝔼{i,j}→{i,j}[βi bi + αj aj, ηi yi + ξj xj, 1]
       (aΔi→1,2 // aΔ2→2,3 // a̅S̅3) (bΔj→-1,-2 // bΔ-2→-2,-3)) // (P-1,3 P-3,1 am2,j→k bmi,-2→k),
  dSi = 𝔼{i}→{1,2}[βi b1 + αi a2, ηi y1 + ξi x2, 1] // (b̅S̅1 aS2) // dm2,1→i,
  dΔi→j,k = (bΔi→3,1 aΔi→2,4) // (dm3,4→k dm1,2→j)]
```

Program

In[∘]:=
```
Define[Ci = 𝔼{}→{i}[0, 0, Bi^(1/2) ⅇ^(-ℏ ϵ ai/2)]$k,
  C̄i = 𝔼{}→{i}[0, 0, Bi^(-1/2) ⅇ^(ℏ ϵ ai/2)]$k,
  Kinki  =  (R1,3 C̄2) // dm1,2→1 // dm1,3→i,
  K̅i̅n̅k̅i  =  (R̄1,3 C2) // dm1,2→1 // dm1,3→i]
```

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[∘]:=
```
Define[b2ti = 𝔼{i}→{i}[αi ai - βi ti / γ, ξi xi + ηi yi, ⅇ^(ϵ βi ai/γ)]$k,
  t2bi = 𝔼{i}→{i}[αi ai - τi γ bi, ξi xi + ηi yi, ⅇ^(ϵ τi ai)]$k]
```

Program

In[∘]:=
```
Define[kRi,j = Ri,j // (b2ti b2tj) /. ti|j → t,
  k̅R̅i,j = R̄i,j // (b2ti b2tj) /. {ti|j → t, Ti|j → T},
  kmi,j→k = (t2bi t2bj) // dmi,j→k // b2tk /. {tk → t, Tk → T, τi|j → 0},
  kCi = Ci // b2ti /. Ti → T, k̅C̅i = C̄i // b2ti /. Ti → T,
  kKinki = Kinki // b2ti /. {ti → t, Ti → T},
  k̅K̅i̅n̅k̅i = K̅i̅n̅k̅i // b2ti /. {ti → t, Ti → T}]
```

Testing

In[ ]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, a̅S̅ → a̅S̅_i, bS → bS_i, b̅S̅ → b̅S̅_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, K̅i̅n̅k̅ → K̅i̅n̅k̅_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$am \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\ (\alpha_i+\alpha_j),\ x_k\ \left(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\ (\beta_i+\beta_j),\ y_k\ (\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}_{\{i,j\}\to\{k\}}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ \frac{1}{\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\left(\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i+\hbar\,y_k\,\mathcal{A}_j\,\eta_j+\hbar\,x_k\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i-B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+\hbar\,x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j\right),$$
$$1+\frac{1}{4\,\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\Big(-4\,\hbar\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j-4\,\hbar\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i+4\,\gamma\,\hbar^2\,x_k\,y_k\,\eta_j\,\xi_i+4\,\hbar\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+2\,\gamma\,\hbar\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i-6\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i+2\,\gamma\,\hbar\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2-$$
$$6\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2+\gamma\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2-4\,\gamma\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2+3\,\gamma\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$R \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1-\frac{\left(4\,\hbar^2\,a_j\,B_i\,x_j\,y_i+3\,\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon}{4\,B_i^2}+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(2\,\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(2\,\gamma\,\hbar\,B_i\,y_i\,\eta_i-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}_{\{i\}\to\{i\}}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ \frac{-\hbar\,y_i\,\mathcal{A}_i\,\eta_i-\hbar\,B_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1+\frac{1}{4\,\hbar\,B_i^2}\Big(4\,\gamma\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,\hbar\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,\gamma\,\hbar^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i-4\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-$$
$$4\,\gamma\,\hbar\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\hbar\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,\gamma\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+$$
$$4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,\gamma\,\hbar\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar^2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+$$
$$6\,\gamma\,\hbar\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\gamma\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$a\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$b\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$d\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,$$
$$y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$C \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2\,\hbar\,a_i-\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\frac{\left(-2\,\hbar\,a_i\,B_i^2-4\,\hbar^2\,a_i\,B_i\,x_i\,y_i-3\,\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,B_i^{3/2}}+O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@
 {{"[a,x]" → ((𝔼{}→{1,2}[0, 0, a₂ x₁] // am1,2→1)[3] - (𝔼{}→{1,2}[0, 0, a₁ x₂] // am1,2→1)[3]),
    "[b,y]" → ((𝔼{}→{1,2}[0, 0, y₂ b₁] // bm1,2→1)[3] - (𝔼{}→{1,2}[0, 0, y₁ b₂] // bm1,2→1)[3])} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼{}→{1}[0, 0, y₁] ~B₁~ bΔ1→1,2],
   "Δ[b]" → Last[𝔼{}→{1}[0, 0, b₁] ~B₁~ bΔ1→1,2],
   "Δ[a]" → Last[𝔼{}→{1}[0, 0, a₁] ~B₁~ aΔ1→1,2],
   "Δ[x]" → Last[𝔼{}→{1}[0, 0, x₁] ~B₁~ aΔ1→1,2]},
  {
    "S(a)" → ((𝔼{}→{1}[0, 0, a₁] ~B₁~ aS₁)[3]),
    "S(x)" → ((𝔼{}→{1}[0, 0, x₁] ~B₁~ aS₁)[3]),
    "S(b)" → ((𝔼{}→{1}[0, 0, b₁] ~B₁~ bS₁)[3]),
    "S(y)" → ((𝔼{}→{1}[0, 0, y₁] ~B₁~ bS₁)[3])
   } /. z_1 → z}
```

Out[∘]= $\{1.0625, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\},$
$\{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,$
$\Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$
$S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
 ]
```

Out[∘]= {0.15625, {True, True}}

R and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[∘]= $\{0.140625, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar a_j b_i,\ \hbar x_j y_i,\ 1 - \frac{1}{4}(\gamma\hbar^3 x_j^2 y_i^2)\epsilon + \left(\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4\right)\epsilon^2 +$
$\frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k\beta_i}{\hbar},\ \frac{\eta_i\xi_k}{\hbar},\ 1 + \frac{\gamma\eta_i^2\xi_k^2\epsilon}{4\hbar} + \frac{(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2}{288\hbar^2} - \frac{1}{1152\hbar^3}$
$(-48\gamma^3\hbar^4\eta_i^2\xi_k^2 - 192\gamma^3\hbar^3\eta_i^3\xi_k^3 - 156\gamma^3\hbar^2\eta_i^4\xi_k^4 - 40\gamma^3\hbar\eta_i^5\xi_k^5 - 3\gamma^3\eta_i^6\xi_k^6)\epsilon^3 + O[\epsilon]^4],\ \text{True}\}\}$

as and $\overline{aS}$ are inverses, bs and $\overline{bS}$ are inverses:

In[∘]:=
```
Timing[HL /@ {(a̅S̅₁ // aS₁) ≡ 𝔼{1}→{1}[a₁ α₁, x₁ ξ₁, 1], (b̅S̅₁ // bS₁) ≡ 𝔼{1}→{1}[b₁ β₁, y₁ η₁, 1]}]
```

Out[∘]= {0.40625, {True, True}}

(co)-associativity on both sides

In[∘]:=
```
Timing[
 HL /@ {(aΔ_{1→1,2} // aΔ_{2→2,3}) ≡ (aΔ_{1→1,3} // aΔ_{1→1,2}), (bΔ_{1→1,2} // bΔ_{2→2,3}) ≡ (bΔ_{1→1,3} // bΔ_{1→1,2}),
   (am_{1,2→1} // am_{1,3→1}) ≡ (am_{2,3→2} // am_{1,2→1}), (bm_{1,2→1} // bm_{1,3→1}) ≡ (bm_{2,3→2} // bm_{1,2→1})}]
```
Out[∘]= {0.453125, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing[HL /@ {(am_{1,2→1} // aΔ_{1→1,2}) ≡ ((aΔ_{1→1,3} aΔ_{2→2,4}) // (am_{3,4→2} am_{1,2→1})),
   (bm_{1,2→1} // bΔ_{1→1,2}) ≡ ((bΔ_{1→1,3} bΔ_{2→2,4}) // (bm_{3,4→2} bm_{1,2→1}))}]
```
Out[∘]= {0.65625, {True, True}}

An explicit formula for aS_i

In[∘]:=
```
Timing@Block[{$k = 4}, HL[aS_i ≡ (𝔼_{{i}→{i,j}}[-α_i a_j, -ξ_i x_i,
      Sum[Expand[(e^{ξ_i x_i} (-ħ γ ϵ)^k)/(2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, e^{-ξ_i e^{ħ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k} //
    am_{i,j→i})]]
```
Out[∘]= {3.48438, True}

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~B_1~ aS_1) ~B_{1,2}~ am_{1,2→1}, (aΔ_{1→1,2} ~B_2~ aS_2) ~B_{1,2}~ am_{1,2→1},
   (bΔ_{1→1,2} ~B_1~ bS_1) ~B_{1,2}~ bm_{1,2→1}, (bΔ_{1→1,2} ~B_2~ bS_2) ~B_{1,2}~ bm_{1,2→1}}]
```
Out[∘]= {0.65625, {True, True, True, True}}

But not with the opposite product:

In[∘]:=
```
Timing[Short[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~B_1~ aS_1) ~B_{1,2}~ am_{2,1→1}, (aΔ_{1→1,2} ~B_2~ aS_2) ~B_{1,2}~ am_{2,1→1},
   (bΔ_{1→1,2} ~B_1~ bS_1) ~B_{1,2}~ bm_{2,1→1}, (bΔ_{1→1,2} ~B_2~ bS_2) ~B_{1,2}~ bm_{2,1→1}}]
```
Out[∘]= $\Big\{0.703125, \Big\{\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \mathcal{A}_1 \xi_1 + \gamma^2\epsilon^2\hbar^2 x_1 \mathcal{A}_1 \xi_1 - 2\gamma \ll 4 \gg \mathcal{A}_1\xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \mathcal{A}_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \xi_1 - \gamma^2\epsilon^2\hbar^2 x_1\xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2\xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar y_1 \eta_1 - \gamma^2\epsilon^2\hbar^2 y_1\eta_1 + 2\gamma^2\epsilon^2\hbar^2 y_1^2\eta_1^2\right) == 0, \frac{-2\gamma\epsilon\hbar B_1 y_1 \eta_1 + \ll 4 \gg}{2 B_1^2} == 0\Big\}\Big\}$

S is an algebra anti-(co)morphism

In[∘]:=
```
Timing[HL /@ {am_{1,2→1} ~B_1~ aS_1 ≡ (aS_1 aS_2) ~B_{1,2}~ am_{2,1→1}, bm_{1,2→1} ~B_1~ bS_1 ≡ (bS_1 bS_2) ~B_{1,2}~ bm_{2,1→1},
   aS_1 ~B_1~ aΔ_{1→1,2} ≡ aΔ_{1→2,1} ~B_{1,2}~ (aS_1 aS_2), bS_1 ~B_1~ bΔ_{1→1,2} ≡ bΔ_{1→2,1} ~B_{1,2}~ (bS_1 bS_2)}]
```
Out[∘]= {0.96875, {True, True, True, True}}

Pairing axioms

In[∘]:=
```
Timing[HL /@ {(bm_{1,2→1} 𝔼_{{3}→{3}}[α_3 a_3, ξ_3 x_3, 1]) ~B_{1,3}~P_{1,3} ≡
    (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] 𝔼_{{2}→{2}}[β_2 b_2, η_2 y_2, 1] aΔ_{3→4,5}) ~B_{1,4}~P_{1,4}~B_{2,5}~P_{2,5},
   (bΔ_{1→1,2} 𝔼_{{3}→{3}}[α_3 a_3, ξ_3 x_3, 1] 𝔼_{{4}→{4}}[α_4 a_4, ξ_4 x_4, 1]) ~B_{1,3}~P_{1,3}~B_{2,4}~P_{2,4} ≡
    (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] am_{3,4→3}) ~B_{1,3}~P_{1,3} }]
```

Out[∘]= {0.484375, {True, True}}

In[∘]:=
```
Timing[HL /@ {((bS_1 𝔼_{{2}→{2}}[α_2 a_2, ξ_2 x_2, 1]) // P_{1,2}) ≡ ((𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] aS_2) // P_{1,2}),
   (overline{bS}_1 𝔼_{{2}→{2}}[α_2 a_2, ξ_2 x_2, 1]) ~B_{1,2}~P_{1,2} ≡ (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] overline{aS}_2) ~B_{1,2}~P_{1,2}}]
```

Out[∘]= {0.296875, {True, True}}

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
Timing@{{
      "[a,y]" →
       ((𝔼_{{}→{1,2}}[0, 0, y_2 a_1] ~B_{1,2}~dm_{1,2→1}) [3] - (𝔼_{{}→{1,2}}[0, 0, y_1 a_2] ~B_{1,2}~dm_{1,2→1}) [3]),
      "[b,x]"  → ((𝔼_{{}→{1,2}}[0, 0, x_2 b_1] ~B_{1,2}~dm_{1,2→1}) [3] -
         (𝔼_{{}→{1,2}}[0, 0, x_1 b_2] ~B_{1,2}~dm_{1,2→1}) [3]),
      "xy-qyx" → ((𝔼_{{}→{1,2}}[0, 0, x_1 y_2] ~B_{1,2}~dm_{1,2→1}) [3] -
         (1 + ϵ) (𝔼_{{}→{1,2}}[0, 0, y_1 x_2] ~B_{1,2}~dm_{1,2→1}) [3])
     } /. {z_1 → z} // Expand // Factor,
   {
    "Δ(a)" → ((𝔼_{{}→{1}}[0, 0, a_1] ~B_1~dΔ_{1→1,2}) [3]),
    "Δ(x)" → ((𝔼_{{}→{1}}[0, 0, x_1] ~B_1~dΔ_{1→1,2}) [3]),
    "Δ(b)" → ((𝔼_{{}→{1}}[0, 0, b_1] ~B_1~dΔ_{1→1,2}) [3]),
    "Δ(y)" → ((𝔼_{{}→{1}}[0, 0, y_1] ~B_1~dΔ_{1→1,2}) [3])
   } // Simplify,
   {
     "S(a)" → ((𝔼_{{}→{1}}[0, 0, a_1] ~B_1~dS_1) [3]),
     "S(x)" → ((𝔼_{{}→{1}}[0, 0, x_1] ~B_1~dS_1) [3]),
     "S(b)" → ((𝔼_{{}→{1}}[0, 0, b_1] ~B_1~dS_1) [3]),
     "S(y)" → ((𝔼_{{}→{1}}[0, 0, y_1] ~B_1~dS_1) [3])
    } /. {z_1 → z} // Simplify
  }
```

Out[∘]= $\{8.04688, \{\{[a,y] \to -y\gamma + O[\epsilon]^3,\ [b,x] \to x\epsilon + O[\epsilon]^3,$

$xy\text{-}qyx \to \left(-xy + \frac{1 - B + xy\hbar}{\hbar}\right) + (aB - xy + xy\gamma\hbar)\,\epsilon + \frac{1}{2}\left(-a^2 B\hbar + xy\gamma^2\hbar^2\right)\epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\},$

$\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - ax\hbar\epsilon - \frac{1}{2}\left(a^2 x\hbar^2\right)\epsilon^2 + O[\epsilon]^3,$

$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\gamma\hbar\epsilon}{B} - \frac{\left(y\gamma^2\hbar^2\right)\epsilon^2}{2B} + O[\epsilon]^3\}\}\}$

(co)-associativity

In[∘]:=
```
Timing[
 HL /@ {(dΔ_{1→1,2} // dΔ_{2→2,3}) ≡ (dΔ_{1→1,3} // dΔ_{1→1,2}), (dm_{1,2→1} // dm_{1,3→1}) ≡ (dm_{2,3→2} // dm_{1,2->1})}]
```
Out[∘]= {7.10938, {True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing@HL[dm_{1,2→1} ~ B_1 ~ dΔ_{1→1,2} ≡ (dΔ_{1→1,3} dΔ_{2→2,4}) ~ B_{1,2,3,4} ~ (dm_{3,4→2} dm_{1,2→1})]
```
Out[∘]= {8.78125, True}

S_2 inverts R, but not S_1:

In[∘]:=
```
Timing@{R_{1,2} ~ B_1 ~ dS_1 ≡ R̄_{1,2}, HL[R_{1,2} ~ B_2 ~ dS_2 ≡ R̄_{1,2}]}
```
Out[∘]= $\left\{0.84375, \left\{\frac{1}{4 B_1^3}\left(4 \gamma \in \hbar^2 B_1^2 x_2 y_1 - 2 \gamma^2 \in^2 \hbar^3 B_1^2 x_2 y_1 + 4 \gamma \in^2 \hbar^3 a_2 B_1^2 x_2 y_1 + 8 \gamma^2 \in^2 \hbar^4 B_1 x_2^2 y_1^2 - 4 \gamma \in^2 \hbar^4 a_2 B_1 x_2^2 y_1^2 - 3 \gamma^2 \in^2 \hbar^5 x_2^3 y_1^3\right) == 0, \text{True}\right\}\right\}$

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@
  {(dΔ_{1→1,2} ~ B_1 ~ dS_1) ~ B_{1,2} ~ dm_{1,2→1}, (dΔ_{1→1,2} ~ B_2 ~ dS_2) // dm_{1,2→1}}]
```
Out[∘]= {10.2969, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:=
```
Timing[HL /@
  Expand /@ {dm_{1,2→1} ~ B_1 ~ dS_1 ≡ (dS_1 dS_2) ~ B_{1,2} ~ dm_{2,1→1}, dS_1 ~ B_1 ~ dΔ_{1→1,2} ≡ dΔ_{1→2,1} ~ B_{1,2} ~ (dS_1 dS_2)}]
```
Out[∘]= {19.2344, {True, True}}

Quasi-triangular axiom 1:

In[∘]:=
```
Timing@HL[R_{1,2} ~ B_1 ~ dΔ_{1->1,3} ≡ (R_{1,4} R_{3,2}) ~ B_{2,4} ~ dm_{2,4→2}]
```
Out[∘]= {0.71875, True}

Quasi-triangular axiom 2:

In[∘]:=
```
Timing@HL[((dΔ_{1->1,2} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2})) ≡ ((dΔ_{1→2,1} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{3,1→1} dm_{4,2→2}))]
```
Out[∘]= {15.0469, True}

The Drinfel'd element inverse property, $(u_1 \bar{u}_2) \sim B_{1,2} \sim \text{dm}_{1,2\to1} \equiv \mathbb{E}[0, 0, 1]$:

In[∘]:=
```
Timing@HL[((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i} ≡
  𝔼_{{}→{i}}[0, 0, 1]]
```
Out[∘]= {6.35938, True}

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

In[∘]:= $\text{Timing@Block}[\{\$k = 2\}, (((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}) \sim B_i \sim dS_i)\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i}]$

Out[∘]= $\left\{8.59375, \mathbb{E}_{\{\}\to\{i\}}\left[0, 0, \frac{1}{B_i} + \frac{\hbar\, a_i\, \epsilon}{B_i} + \frac{\hbar^2 a_i^2 \epsilon^2}{2 B_i} + O[\epsilon]^3\right]\right\}$

In[∘]:= $\text{Timing@Block}[\{\$k = 2\}, \text{HL /@}\{(C_i \bar{C}_j) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}_{\{\}\to\{i\}}[0, 0, 1], (\bar{C}_i \bar{C}_j) \sim B_{i,j} \sim dm_{i,j\to i} \equiv (((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}) \sim B_i \sim dS_i)\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i}\}]$

Out[∘]= {8.92188, {True, True}}

Reidemeister 2:

In[∘]:= $\text{Timing}[\text{HL}[\# \equiv \mathbb{E}_{\{\}\to\{1,2\}}[0, 0, 1]]\ \&\text{/@} \{(\bar{R}_{1,2} R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to 1}\, dm_{2,4\to 2}), (R_{1,2} \bar{R}_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to 1}\, dm_{2,4\to 2})\}]$

Out[∘]= {11.1406, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= $\text{Timing@HL}[(R_{1,4} \bar{R}_{5,2} \bar{C}_3) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \bar{C}_1\, \mathbb{E}_{\{\}\to\{2\}}[0, 0, 1]]$

Out[∘]= {4.23438, True}

Reidemeister 3:

In[∘]:= $\text{Timing@HL}[((R_{1,2} R_{4,3} R_{5,6}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}) \equiv ((R_{1,6} R_{2,3} R_{4,5}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3})]$

Out[∘]= {8.73438, True}

Relations between the four kinks:

In[∘]:= $\text{Timing}[\text{HL /@}\{\text{Kink}_i \equiv (R_{3,1} C_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i}, \overline{\text{Kink}}_j \equiv (\bar{R}_{3,1} \bar{C}_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}, (\text{Kink}_i\, \overline{\text{Kink}}_j) \sim B_{i,j} \sim dm_{i,j\to 1} \equiv \mathbb{E}_{\{\}\to\{1\}}[0, 0, 1]\}]$

Out[∘]= {7.75, {True, True, True}}

Trefoil

The Trefoil

Trefoil

In[∘]:=
```
$k = 2; Z = kR1,5 kR6,2 kR3,7 kC4 kKink8 kKink9 kKink10;
Do[Z = Z ~ B1,r ~ km1,r→1, {r, 2, 10}];
Simplify /@ Z /. ν_1 :→ ν
```

Trefoil

Out[∘]=
$$\mathbb{E}_{\{\}\to\{1\}}\Big[0,\ 0,\ \frac{T}{1-T+T^2}+\Big(T\hbar\left(2a\left(-1+T-T^3+T^4\right)+T\left(-1+2T-3T^2+2T^3\right)\gamma-2\left(1+T^3\right)xy\gamma\hbar\right)\epsilon\Big)\Big/\left(1-T+T^2\right)^3+\frac{1}{2\left(1-T+T^2\right)^5}T\hbar^2\Big(4a^2\left(1-T+T^2\right)^2\left(1+T-6T^2+T^3+T^4\right)+4a\left(1-T+T^2\right)\gamma\left(T\left(2-5T+8T^2-7T^3-2T^4+2T^5\right)-2\left(-1-2T+5T^2-4T^3+T^4+2T^5\right)xy\hbar\right)+\gamma^2\Big(T\left(1-2T+4T^2-2T^3+6T^5-11T^6+4T^7\right)+4\left(-1+2T+T^3+T^4+2T^6-T^7\right)xy\hbar+6\left(1-T+T^2\right)^2\left(1+3T+T^2\right)x^2y^2\hbar^2\Big)\Big)\epsilon^2+O[\epsilon]^3\Big]$$